

Basic Physics

Physics Examples from Napolitano Mathematica Primer for Physicists

```
In[46]:= Remove["Global`*"]
```

Chapter 1 Introduction

Example 1.1

A resistor R and capacitor C are arranged in series with a switch and a source of EMF V . The capacitor is initially uncharged when the switch is closed, so the charge Q on the capacitor, as a function of time t , is $Q(t) = CV(1 - e^{-t/\tau})$ where $\tau = RC$.

For $V = 1$ Volt and $C = 1\mu\text{F}$ (1 farad = 10^6 microfarads), plot $Q(t)$ for $t = 0$ to $t = 5$ ms for each of the three values $R = 1\text{ k}\Omega$, $R = 2\text{ k}\Omega$, and $R = 5\text{ k}\Omega$.

```
In[1]:= (*CHARGING A CAPACITOR *)
```

```
(*Define the equation. Write out the formulae in their most abstract forms.*)
```

```
 $\tau = r c;$ 
```

```
 $q = c v (1 - \text{Exp}[-t / \tau]);$ 
```

```
(*Form expressions corresponding to each R.*)
```

```
 $c = 10^{-6};$ 
```

```
 $v = 1;$ 
```


```
 $q1 = q /. r \rightarrow 1 * 10^3;$ 
```

```
 $q2 = q /. r \rightarrow 2 * 10^3;$ 
```

```
 $q3 = q /. r \rightarrow 5 * 10^3;$ 
```

```
In[105]:= (*Plot the three eqns with scaling.*)
qVals = 10^6 {q1, q2, q3} /. t -> 10^(-3) tms;
```

```
Plot[qVals, {tms, 0, 5},
PlotStyle -> {Blue, {Brown, Dashed}, {Black, Dotted}},
PlotLegends -> {"1kΩ", "2kΩ", "5kΩ"},
AxesLabel -> {"t (ms)", "Q (μC)"}
]
```



Chapter 2 Algebraic Equations

Example 2.1

Train #1 and train #2 move side-by-side on parallel horizontal straight tracks. At time $t = 0$, train #1 starts from rest and accelerates at a constant rate $a_1 = 0.5 \text{ m/s}^2$. Also at $t = 0$, train #2 passes train #1 while moving at 20 m/s , in the same direction as train #1 is accelerating, but decelerating at a rate $a_2 = 0.2 \text{ m/s}^2$. Find the time at which train #1 passes train #2, and determine the distance from the start at which they pass. Set things up so that position $x = 0$ locates the trains at $t = 0$. That is, $x_1 = \frac{1}{2}a_1t^2$ --- (2.1a), $x_2 = v_0t - \frac{1}{2}a_2t^2$ --- (2.1b) where $a_1 = 0.5$, $a_2 = 0.2$, and $v_0 = 20$. We need to find t when $x_1 = x_2$.

```
In[2]:= (* PARALLEL TRAINS IN ONE DIMENSION *)
```

```
(*Define all the equations and solve them.*)
```

$$x1 = 1/2 a1 t^2$$

$$x2 = v0 t - 1/2 a2 t^2$$

$$\text{Out[2]} = \frac{a1 t^2}{2}$$

$$\text{Out[3]} = -\frac{a2 t^2}{2} + t v0$$

```
In[12]:= soln = Solve[x1 == x2, t]
```

```
tFinite = Part[Part[soln, 2], 1]
```

$$\text{Out[12]} = \left\{ \{t \rightarrow 0\}, \left\{ t \rightarrow \frac{2 v0}{a1 + a2} \right\} \right\}$$

$$\text{Out[13]} = t \rightarrow \frac{2 v0}{a1 + a2}$$

```
In[6]:= (*Check the solutions.*)
```

```
x1Meet = x1 /. tFinite
```

$$\text{Out[6]} = \frac{2 a1 v0^2}{(a1 + a2)^2}$$

```
In[7]:= x2Meet = x2 /. tFinite // Simplify
```

$$\text{Out[7]} = \frac{2 a1 v0^2}{(a1 + a2)^2}$$

```
In[8]:= (*Input numbers.*)
```

```
vals = {a1 -> 0.5, a2 -> 0.2, v0 -> 20};
```

```
x1Meet /. vals
```

```
Out[9]= 816.327
```

```
In[11]:= Print["The two trains meet at ", x1Meet /. vals,
```

```
" meters, which is after ", t /. tFinite /. vals, " seconds."]
```

The two trains meet at 816.327 meters, which is after 57.1429 seconds .

Example 2.2

Figure 2.1 shows a DC circuit. Use Kirchoff's Laws to determine the currents i_1 , i_2 , and i_3 in terms of V , R_1 , R_2 , and R_3 . Kirchoff's node and loop laws give the following equations: $i_1 = i_2 + i_3$ (2.2a); $V = i_1 R_1 + i_2 R_2$ (2.2b); $0 = i_2 R_2 - i_3 R_3$ (2.2c); $V = i_1 R_1 + i_3 R_3$ (2.2d).

```
(* CURRENT IN A DC CIRCUIT *)
(*Define the eqns and solve.*)
eq1 = i1 == i2 + i3;
eq2 = v == i1 r1 + i2 r2;
eq3 = 0 == i2 r2 - i3 r3;
eq4 = v == i1 r1 + i3 r3;
```

```
Solve[{eq1, eq2, eq3, eq4}, {i1, i2, i3}]
```

Out[460]= $\left\{ \left\{ i1 \rightarrow -\frac{-r_2 v - r_3 v}{r_1 r_2 + r_1 r_3 + r_2 r_3}, i2 \rightarrow \frac{r_3 v}{r_1 r_2 + r_1 r_3 + r_2 r_3}, i3 \rightarrow \frac{r_2 v}{r_1 r_2 + r_1 r_3 + r_2 r_3} \right\} \right\}$

Example 2.3

Three point charges are arranged in a straight line. Charges $q_1=1\text{C}$, $q_2=-2\text{C}$, and $q_3=4\text{C}$ are located at $x_1=-2\text{m}$, $x_2=0$, and $x_3=1\text{m}$, respectively. Find the point or points along the line where the electric field is zero.

```
(* FINDING ZEROS OF AN ELECTRIC FIELD *)
```

```
e[i_] := qs[[i]]/((x - xs[[i]])^2);(*This comes from the fact  $E_i = Q_i / (x - x_i^2)$ *)
```

```
qs = {1, -2, 4};
```

```
xs = {-2, 0, 1};
```


```
eee = e[1] + e[2] + e[3]
```

Out[31]=
$$\frac{4}{(-1+x)^2} - \frac{2}{x^2} + \frac{1}{(2+x)^2}$$

```
In[32]:= Solve[eee == 0, x]
```

Out[32]= $\left\{ \{x \rightarrow -1\}, \{x \rightarrow 0.412 \dots\}, \{x \rightarrow -1.37 \dots - 2.14 \dots i\}, \{x \rightarrow -1.37 \dots + 2.14 \dots i\} \right\}$

```
In[35]:= Plot[eee, {x, -2, 2}, AxesLabel -> {"x: distance", "eee: electric field"}]
```



Chapter 3 Derivatives Integrals Series

Example 3.1

A particle of mass m moves in one dimension x according to $x(t) = A e^{-\beta t} \cos \omega t$.

Show that this motion corresponds to the mass being acted on by a restoring force $-kx$ and a damping force $-bv$, where v is the velocity, $\omega = \sqrt{k/m}$ and $\beta = b/2m$.

```
In[58]:= (*THE DAMPED OSCILLATOR*)

(*Construct distance, velocity, and acceleration functions.*)
x = A Exp[- $\beta$  t]  $\times$  Cos[ $\omega$  t]
v = D[x, t]
a = D[v, t]

Out[58]= A e- $\beta$  t Cos[t  $\omega$ ]

Out[59]= -A e- $\beta$  t  $\beta$  Cos[t  $\omega$ ] - A e- $\beta$  t  $\omega$  Sin[t  $\omega$ ]

Out[60]= A e- $\beta$  t  $\beta^2$  Cos[t  $\omega$ ] - A e- $\beta$  t  $\omega^2$  Cos[t  $\omega$ ] + 2 A e- $\beta$  t  $\beta \omega$  Sin[t  $\omega$ ]

In[61]:= (*Construct force function and compare it to mass x acceleration.*)
f = -k x - b v // Simplify

Out[61]= A e- $\beta$  t ((-k + b  $\beta$ ) Cos[t  $\omega$ ] + b  $\omega$  Sin[t  $\omega$ ])

In[62]:= f /. {k  $\rightarrow$  m ( $\omega^2$ ), b  $\rightarrow$  2 m  $\beta$ } // Simplify

Out[62]= A e- $\beta$  t m ((2  $\beta^2$  -  $\omega^2$ ) Cos[t  $\omega$ ] + 2  $\beta \omega$  Sin[t  $\omega$ ])

In[63]:= m a // Simplify

Out[63]= A e- $\beta$  t m (( $\beta^2$  -  $\omega^2$ ) Cos[t  $\omega$ ] + 2  $\beta \omega$  Sin[t  $\omega$ ])

(* f and ma should be equal. There is an inaccuracy in the above presentation. *)
```

Example 3.2

Consider a pendulum made from a massless string of length l and a bob of mass m . We generally write the period as $2\pi\sqrt{l/g}$ but this is accurate only if the maximum displacement angle θ_0 much less than 1. Find an expression(perhaps in terms of a special function) for the period as a function of θ_0 . Form the ratio of the period to $2\pi\sqrt{l/g}$ and plot it for $0 \leq \theta_0 \leq 0.99\pi$.

```
In[110]:= (* PENDULUM PERIOD *)

(*Integrate to get the period.*)
t1 = AbsoluteTime [];
int = Integrate[1 / Sqrt[Cos[ $\theta$ ] - Cos[ $\theta_0$ ]], { $\theta$ ,  $\theta_0$ , 0},
Assumptions  $\rightarrow$  { $\theta_0 > 0$ ,  $\theta_0 < \text{Pi}$ }]
t2 = AbsoluteTime [];
t2 - t1

Out[111]= -  $\frac{2 \text{EllipticF}\left[\frac{\theta_0}{2}, \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right]}{\sqrt{1 - \text{Cos}[\theta_0]}}$ 


Out[113]= 2.627893
```

```
In[109]:= period = -4 Sqrt[1 / (2 g)] int
```

$$\text{Out[109]} = \frac{4 \sqrt{2} \sqrt{\frac{1}{g}} \text{EllipticF}\left[\frac{\theta_0}{2}, \text{Csc}\left[\frac{\theta_0}{2}\right]^2\right]}{\sqrt{1 - \text{Cos}[\theta_0]}}$$

```
In[120]:= ratio = period / (2 Pi Sqrt[1/g]);
```

```
LogPlot[ratio, {\theta_0, 0, .99 Pi}]
```



Example 3.3

At the Earth's surface, we speak of the “acceleration g due to gravity”. In terms of Newton's theory and a spherical Earth of radius R and mass M , $g = GM/R^2$. Find an expression up to second order for the modification to g for an object at height h .

(* GRAVITY NEAR THE SURFACE OF EARTH *)

```
In[16]:= (*Define the force of gravity.*)
```

```
force = G m M / (R + h)^2
```

$$\text{Out[16]} = \frac{G m M}{(h + R)^2}$$

(*Substitute and expand.*)

```
Series[force /. h -> x R, {x, 0, 2}]
```

(*Series approximation of function force around $x=0$ up to order 2*)

$$\text{Out[17]} = \frac{G m M}{R^2} - \frac{2 (G m M) x}{R^2} + \frac{3 G m M x^2}{R^2} + O[x]^3$$

```
In[18]:= forcex = Normal[%] /. x -> h / R
```

$$\text{Out[18]} = \frac{3 G h^2 m M}{R^4} - \frac{2 G h m M}{R^3} + \frac{G m M}{R^2}$$

```
In[19]:= grepl = Solve[g == G M / R^2, G]
```

$$\text{Out[19]} = \left\{ \left\{ G \rightarrow \frac{g R^2}{M} \right\} \right\}$$

In[29]:= `forcex /. grepl [[1]] // Simplify`

Out[29]=
$$\frac{g m (3 h^2 - 2 h R + R^2)}{R^2}$$

Chapter 4 Differential Equations Analytical Solutions

Example 4.1

A mass m is fired vertically upward from the Earth's surface with an initial speed v_0 . The mass is subject to a drag force bv , that is, proportional to its velocity. Find an expression for the time it takes the projectile to reach its highest point. For $m = 100\text{g}$ and $v_0 = 20\text{m/s}$, plot this time versus the drag coefficient b for $0 \leq b \leq 1$. Confirm your value for t when $b = 0$.

(* VERTICAL PROJECTILE WITH DRAG *)

(*Solve the differential eq.*)

`sol = DSolve[{m v'[t] == -m g - b v[t], v[0] == v0}, v, t]`

(* Solve diffeqs with respect to v with independent var t*)

Out[38]=
$$\left\{ \left\{ v \rightarrow \text{Function}[\{t\}, -\frac{e^{-\frac{b t}{m}} \left(-g m + e^{\frac{b t}{m}} g m - b v_0 \right)}{b} \right] \right\} \right\}$$

In[34]:= `solAlone = Part[Part[sol, 1], 1]`

Out[34]=
$$v \rightarrow \text{Function}[\{t\}, -\frac{e^{-\frac{b t}{m}} \left(-g m + e^{\frac{b t}{m}} g m - b v_0 \right)}{b}]$$


In[50]:= `vel = v[t] /. solAlone`

Out[50]=
$$-\frac{e^{-\frac{b t}{m}} \left(-g m + e^{\frac{b t}{m}} g m - b v_0 \right)}{b}$$

In[48]:= `solt = Solve[vel == 0, t, Reals] // Expand`

Out[48]=
$$\left\{ \left\{ t \rightarrow \frac{m \operatorname{Log}\left[\frac{g m + b v_0}{g m}\right]}{b} \text{ if } \text{condition} \right\} \right\}$$

```
In[53]:= (*Make plot with specific values.*)
vals = {m → 0.1, g → 9.8, v0 → 20};
Plot[t /. solt /. vals, {b, 0, 1},
PlotRange → {{0, 1}, {0, 2.5}}
```



```
In[55]:= (*Check for b=0.*)
v0 / g /. vals
```

Out[55]= 2.04082

Example 4.2

A mass $m = 100\text{g}$, initially at rest, is dropped from a height $h = 100\text{m}$ and is subject to a linear drag force bv , proportional to its velocity. Find an expression for its height as a function of time, and plot it until it hits the ground for various values of b . Comment on the shape of the curves.

(* A FALLING BALL WITH DRAG *)

```
In[149]:= (* Solve the differential equation. *)
DSolve[{m y'[t] == - m g - b y'[t],
        y[0] == h,
        y'[0] == 0}, y, t];
```

```
height = y[t] /. % [[1]] // Simplify (*This is inserting the solution from
the previous line into y[t] to derive the expression for the height. *)
```

Out[150]=

$$h + \frac{g m \left(m - e^{-\frac{b t}{m}} m - b t \right)}{b^2}$$

(* Check limit as $b \rightarrow 0$ *)


```
In[151]:= Limit[height, b → 0]
```

```
Out[151]= h -  $\frac{g t^2}{2}$ 
```


```
In[152]:= (*Plot*)
```

```
mgh = {m → 0.1, g → 9.8, h → 100};
```

```
Manipulate[Plot[height /. mgh /. b → drag, {t, 0, 15}, PlotRange → {0, 100}],
```

```
{drag, 0.0001, 1}]
```

```
Out[153]=
```



Example 4.3

Two masses m are connected by three springs each with stiffness k to each other and to fixed walls on either side of a frictionless, horizontal surface. See Figure 4.1. The two masses are initially at rest, with the mass on the left displaced by a distance A and the other mass at its equilibrium position. Find and plot the positions of each mass as a function of time, in units of $\tau \equiv 2\pi(m/k)^{1/2}$. Also plot the sum and difference of their positions.

```
In[13]:= (* COUPLED MASS AND STRING OSCILLATIONS *)
```

```
(* Set up and solve coupled equations. *)
```

```
eq1 := m x1''[t] == -k x1[t] + k (x2[t] - x1[t]);
```

```
eq2 := m x2''[t] == -k x2[t] - k (x2[t] - x1[t]);
```

```
sol = DSolve[{eq1, eq2, x1[0] == A, x1'[0] == 0, x2[0] == 0, x2'[0] == 0}, {x1, x2}, t]
```

```
Out[15]= {{x1 → Function[{t},  $\frac{1}{4} A e^{-\frac{i \sqrt{k} t}{\sqrt{m}} - \frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} \left( e^{\frac{i \sqrt{k} t}{\sqrt{m}}} + e^{\frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} + e^{\frac{2 i \sqrt{k} t}{\sqrt{m}} + \frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} + e^{\frac{i \sqrt{k} t}{\sqrt{m}} + \frac{2 i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} \right)$ ],
```

$$x2 \rightarrow \text{Function}[\{t\}, -\frac{1}{4} A e^{-\frac{i \sqrt{k} t}{\sqrt{m}} - \frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} \left(e^{\frac{i \sqrt{k} t}{\sqrt{m}}} - e^{\frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} - e^{\frac{2 i \sqrt{k} t}{\sqrt{m}} + \frac{i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} + e^{\frac{i \sqrt{k} t}{\sqrt{m}} + \frac{2 i \sqrt{3} \sqrt{k} t}{\sqrt{m}}} \right)]}}$$

```

In[16]:= (* Extract the solutions. *)
x1sol = x1[t] /. Part[Part[sol, 1], 1]
x2sol =
  x2[t] /. Part[Part[sol, 1], 2] (* Do not Simplify here; it will mess up the Plot. *)

Out[16]= 
$$\frac{1}{4} A e^{-\frac{i\sqrt{k}t}{\sqrt{m}} - \frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} \left( e^{\frac{i\sqrt{k}t}{\sqrt{m}}} + e^{\frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} + e^{\frac{2i\sqrt{k}t}{\sqrt{m}} + \frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} + e^{\frac{i\sqrt{k}t}{\sqrt{m}} + \frac{2i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} \right)$$



Out[17]= 
$$-\frac{1}{4} A e^{-\frac{i\sqrt{k}t}{\sqrt{m}} - \frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} \left( e^{\frac{i\sqrt{k}t}{\sqrt{m}}} - e^{\frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} - e^{\frac{2i\sqrt{k}t}{\sqrt{m}} + \frac{i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} + e^{\frac{i\sqrt{k}t}{\sqrt{m}} + \frac{2i\sqrt{3}\sqrt{k}t}{\sqrt{m}}} \right)$$


In[25]:= (* Massage the solutions so they can be scaled. *)
r = 2 Pi Sqrt[m/k];
subs = {A → 5, t → tScale r};
$Assumptions = m > 0 && k > 0;
x1plot = Simplify[x1sol /. subs];
x2plot = Simplify[x2sol /. subs];

ExpToTrig[x1plot];
ExpToTrig[x2plot];

```

```
In[32]:= (*Make the plots. *)
Plot[{x1plot, x2plot}, {tScale, 0, 2},
  PlotStyle -> {Solid, Dashed}, PlotLegends -> {x1, x2}]
Plot[{x1plot + x2plot, x1plot - x2plot}, {tScale, 0, 2},
  PlotStyle -> {Solid, Dashed}, PlotLegends -> {x1 + x2, x1 - x2}]
```



```
In[299]:=
```

Chapter 5 Differential Equations Numerical Solutions

Example 5.1

An object of mass m moves in one dimension x under a force $F(x) = ax^2 - bx$, where a and b are positive constants. Solve for $x(t)$ with initial conditions $\dot{x}(0) = 0$ and $x(0) = x_0$, and plot the results, for $0 \leq t \leq t_{\text{Max}}$ where t_{Max} is large enough for you to see the behavior as $t \rightarrow \infty$. Let x_0 take on each of three values, with (a) $|x_0| \ll b/2a$, (b) $x_0 > -b/2a$ (by a small amount), and (c) $x_0 < -b/2a$ (also by a small amount). Find a physical reason to explain why $x = -b/2a$ is special.

```
(*A HUMPY POTENTIAL WELL*)
(* Set up eqn, initial conditions, and max time. Take a=1 and b=2.*)
diffeq = D[D[f[u], u], u] == f[u]^2 - 2 f[u]
y0a = 0.01;
y0b = -0.99;
y0c = -1.01;
ica = {f'[0] == 0, f[0] == y0a};
icb = {f'[0] == 0, f[0] == y0b};
icc = {f'[0] == 0, f[0] == y0c};
umax = 10;
```

Out[43]= $f''[u] == -2 f[u] + f[u]^2$

```
In[57]:= (* Solve the eq for 3 initial conditions. *)
sol = NDSolve[{diffeq, ica}, f[u], {u, 0, umax}]
(*solve {eqns} wrt f[u] with independent var u*)
ya = f[u] /. Part[Part[sol, 1], 1]
sol = NDSolve[{diffeq, icb}, f[u], {u, 0, umax}]
yb = f[u] /. Part[Part[sol, 1], 1]
sol = NDSolve[{diffeq, icc}, f[u], {u, 0, umax}]
yc = f[u] /. Part[Part[sol, 1], 1]
```

Out[57]= $\left\{ \left\{ f[u] \rightarrow \text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 10.\} \} \\ \text{Output : scalar} \end{array} \right] [u] \right\} \right\}$

Out[58]= $\text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 10.\} \} \\ \text{Output : scalar} \end{array} \right] [u]$

Out[59]= $\left\{ \left\{ f[u] \rightarrow \text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 10.\} \} \\ \text{Output : scalar} \end{array} \right] [u] \right\} \right\}$




Out[60]= $\text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 10.\} \} \\ \text{Output : scalar} \end{array} \right] [u]$

NDSolve : At u == 6.95901 , step size is effectively zero ; singularity or stiff system suspected .

Out[61]= $\left\{ \left\{ f[u] \rightarrow \text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 6.96\} \} \\ \text{Output : scalar} \end{array} \right] [u] \right\} \right\}$


Out[62]= $\text{InterpolatingFunction} \left[\begin{array}{c} \text{Domain : } \{ \{0., 6.96\} \} \\ \text{Output : scalar} \end{array} \right] [u]$

```
In[63]:= (*Plot the three cases.*)  
Plot[ya, {u, 0, umax}]  
Plot[yb, {u, 0, umax}]  
Plot[yc, {u, 0, umax}, PlotRange -> {-1, 10}]
```



```
In[67]:= (*Consider the Potential Energy. *)
poten = -Integrate[ξ^2 - 2 ξ, {ξ, 0, y}]
Plot[poten, {y, -1.1, 3.5}]
```

Out[67]= $y^2 - \frac{y^3}{3}$



Example 5.2

A long straight thin metal rod has length l . Its temperature $T=T(x,t)$ at time t , where x measures position along the rod, is governed by $\partial T/\partial t = \kappa \partial^2 T/\partial x^2$ (5.6) and the thermal diffusivity constant κ is a property of the metal. The rod starts out at a uniform temperature $T(x,0)=T_0$, but with one end in contact with an ice batch at $T=0^\circ\text{C}$ and the other in boiling water at $T=100^\circ\text{C}$. Determine and plot $T(x,t)$ for different values of T_0 . Show the temperature distribution both for “short” and “long” times. (What are the natural length and time scales?)

```
In[92]:= (* HEAT TRANSFER IN ONE DIMENSION *)
(* Set up and solve the eqns for different times. *)
pde = D[T[y, u], u] == D[D[T[y, u], y], y];
bcs = {T[0, u] == 0, T[1, u] == 100, T[y, 0] == T0}; (*boundary conditions*)
T0a = 50;
T0b = 150;
```

```
sola = NDSolve[{pde, bcs /. T0 -> T0a}, T[y, u], {y, 0, 1}, {u, 0, 10}];
solb = NDSolve[{pde, bcs /. T0 -> T0b}, T[y, u], {y, 0, 1}, {u, 0, 10}];
```


NDSolve : Warning : boundary and initial conditions are inconsistent .

NDSolve : Warning : boundary and initial conditions are inconsistent .

```
In[135]:= (* Plots for 2 different initial temperatures . *)
tempS = T[y, u] /. solA /. u -> 0.1;
tempM = T[y, u] /. solA /. u -> 1;
tempL = T[y, u] /. solA /. u -> 10;
pl1 = Plot[{tempS, tempM, tempL}, {y, 0, 1}, PlotStyle -> {Solid, Dotted, Dashed},
PlotRange -> {0, 150}, PlotLegends -> {"time u = 0.1", "time u = 1", "time u = 10"},
AxesLabel -> {"time", "temperature"}, PlotLabel -> "T0 = 50"];
```

```
In[139]:= tempS = T[y, u] /. solb /. u -> 0.1;
tempM = T[y, u] /. solb /. u -> 1;
tempL = T[y, u] /. solb /. u -> 10;
pl2 = Plot[{tempS, tempM, tempL}, {y, 0, 1},
PlotStyle -> {Solid, Dotted, Dashed}, PlotRange -> {0, 150},
PlotLegends -> {"time u = 0.1", "time u = 1", "time u = 10"},
AxesLabel -> {"time", "temperature"}, PlotLabel -> "T0 = 150"];
```

```
In[147]:= {{pl1}, {pl2}}
```



```
In[149]:=
```

```
In[154]:=
```

```
In[155]:=
```

Chapter 6 Linear Algebra

Example 6.1

Four masses m lie in the xy plane at the corners of a rectangle as shown in Figure 6.1, with one of the masses located at the origin. Find the location of the center of mass, and the principal axes and moments of inertia, about axes located at the center of mass.

```

(* FOUR MASSES IN A PLANE. *)

(* Calculate the center of mass. *)

In[51]:= rvecs = {{0, 0}, {a, 0}, {0, b}, {a, b}};
mvals = {m, m, m, m};
mr = mvals rvecs;
mtot = Total@mvals;
{xCM, yCM} = Total[mr] / mtot

Out[55]=  $\left\{\frac{a}{2}, \frac{b}{2}\right\}$ 

In[78]:= (* Form Inertia Tensor relative to CM. *)
rCM = rvecs - Table[{xCM, yCM}, 4]
mrCM = mvals rCM
mrCM2 = mvals rCM^2
iterm1 = Total@Total@mrCM2
iterm2 = Sum[KroneckerProduct[Part[mrCM, alpha], Part[rCM, alpha]],
             {alpha, 1, 4}]
inertia = iterm1 IdentityMatrix[2] - iterm2

Out[78]=  $\left\{\left\{-\frac{a}{2}, -\frac{b}{2}\right\}, \left\{\frac{a}{2}, -\frac{b}{2}\right\}, \left\{-\frac{a}{2}, \frac{b}{2}\right\}, \left\{\frac{a}{2}, \frac{b}{2}\right\}\right\}$ 

Out[79]=  $\left\{\left\{-\frac{a m}{2}, -\frac{b m}{2}\right\}, \left\{\frac{a m}{2}, -\frac{b m}{2}\right\}, \left\{-\frac{a m}{2}, \frac{b m}{2}\right\}, \left\{\frac{a m}{2}, \frac{b m}{2}\right\}\right\}$ 

Out[80]=  $\left\{\left\{\frac{a^2 m}{4}, \frac{b^2 m}{4}\right\}, \left\{\frac{a^2 m}{4}, \frac{b^2 m}{4}\right\}, \left\{\frac{a^2 m}{4}, \frac{b^2 m}{4}\right\}, \left\{\frac{a^2 m}{4}, \frac{b^2 m}{4}\right\}\right\}$ 

Out[81]=  $a^2 m + b^2 m$ 

Out[82]=  $\{\{a^2 m, 0\}, \{0, b^2 m\}\}$ 

Out[83]=  $\{\{b^2 m, 0\}, \{0, a^2 m\}\}$ 

In[84]:= (* Get eigenvalues and eigenvectors. *)
Eigensystem[inertia]

Out[84]=  $\{\{a^2 m, b^2 m\}, \{\{0, 1\}, \{1, 0\}\}\}$ 

```

Example 6.2 Skipped

Example 6.3

Two masses m_1 and m_2 are connected by three springs with stiffness k_1 , k_2 , and k_3 to each other and to fixed walls on either side of a frictionless, horizontal surface, as shown in Figure 6.2. Find the eigenfrequencies and eigenmodes of oscillation. Analyze the special case $m_1=m_2$, $k_1=k_3$, and $k_2=rk_1$. Show that your answer agrees with the special case in Exercise 6.2 when $r=1$. Discuss the solutions for $r \neq 1$ and $r \ll 1$.


```

In[125]:= (* GENERAL 2M 3K PROBLEM *)
(* Define the matrices. *)
kM = {{k1 + k2, -k2}, {-k2, k2 + k3}};
mM = {{m1, 0}, {0, m2}};

In[120]:= (* Find eigenvalues and eigenvectors. *)
{vals, vecs} = Eigensystem[{kM, mM}]

Out[120]= 
$$\left\{ \frac{k_2 m_1 + k_3 m_1 + k_1 m_2 + k_2 m_2 - \sqrt{-4 (k_1 k_2 + k_1 k_3 + k_2 k_3) m_1 m_2 + (-k_2 m_1 - k_3 m_1 - k_1 m_2 - k_2 m_2)^2}}{2 m_1 m_2}, \right.$$


$$\left. \frac{k_2 m_1 + k_3 m_1 + k_1 m_2 + k_2 m_2 + \sqrt{-4 (k_1 k_2 + k_1 k_3 + k_2 k_3) m_1 m_2 + (-k_2 m_1 - k_3 m_1 - k_1 m_2 - k_2 m_2)^2}}{2 m_1 m_2} \right\}$$


$$, \left\{ \left\{ -\frac{1}{2 k_2 m_1} \left( -k_2 m_1 - k_3 m_1 + k_1 m_2 + k_2 m_2 - \sqrt{-4 (k_1 k_2 + k_1 k_3 + k_2 k_3) m_1 m_2 + (-k_2 m_1 - k_3 m_1 - k_1 m_2 - k_2 m_2)^2} \right), \right.

$$\left. 1 \right\}, \left\{ -\frac{1}{2 k_2 m_1} \left( -k_2 m_1 - k_3 m_1 + k_1 m_2 + k_2 m_2 + \sqrt{-4 (k_1 k_2 + k_1 k_3 + k_2 k_3) m_1 m_2 + (-k_2 m_1 - k_3 m_1 - k_1 m_2 - k_2 m_2)^2} \right), 1 \right\} \right\}$$


In[121]:= (* Check the answers. *)
rep = {m1 → m, m2 → m, k1 → k, k2 → r k, k3 → k};
assmp = {k > 0, m > 0, r > 0}
Simplify[vals /. rep, Assumptions → assmp]
Simplify[vecs /. rep, Assumptions → assmp]

Out[122]= {k > 0, m > 0, r > 0}

Out[123]=  $\left\{ \frac{k}{m}, \frac{k + 2 k r}{m} \right\}$ 

Out[124]= {{1, 1}, {-1, 1}}$$

```

Chapter 7 Data Analysis

Example 7.1 Requires loading data. Done in Jupyter Notebook.

Chapter 8 Fitting Data

Example 8.1 Jupyter Notebook.

Chapter 9 Numerical Manipulations

Example 9.1

Calculate the nominal value of the acceleration g due to gravity at the Earth's surface, from the Earth's mass and nominal radius.

```
In[29]:= (* GRAVITATIONAL ACCELERATION ON EARTH'S SURFACE *)
(* Set up and solve the basic eqn. *)
sol = Solve[m g == G M m / R^2, g];
g = g /. sol[[1]]

Out[30]= 
$$\frac{G M}{R^2}$$

```

```
In[31]:= (* Put in numbers. *)
G = Quantity["GravitationalConstant "] // UnitConvert // QuantityMagnitude
M = Quantity["EarthMass "] // UnitConvert // QuantityMagnitude
R = Quantity["EarthMeanRadius "] // UnitConvert // QuantityMagnitude
g
```

```
Out[31]=  $6.674 \times 10^{-11}$ 
```

```
Out[32]=  $5.97 \times 10^{24}$ 
```

```
Out[33]=  $6.37101 \times 10^6$ 
```

```
Out[34]= 9.82028
```

Example 9.2

A nuclear reactor gets its energy mainly from the fission of the ^{235}U nucleus after it absorbs a very low energy neutron. A typical reaction is $n + ^{235}\text{U} \rightarrow ^{92}\text{Kr} + ^{144}\text{Ba}$. Calculate the number of fissions per second in a 500MW reactor. How long does it take to burn up 10 kg of ^{235}U ?

```
In[59]:= (* ENERGY FROM  $^{235}\text{U}$  FISSION *)
(* Get the isotope data. *)

c = QuantityMagnitude @ UnitConvert @ Quantity @ "SpeedOfLight "
m236 = QuantityMagnitude @ UnitConvert @ IsotopeData["Uranium236 ", "AtomicMass "]
E236 = m236 c ^ 2
E092 =
  (QuantityMagnitude @ UnitConvert @ IsotopeData["Krypton92 ", "AtomicMass "]) c ^ 2
E144 = (QuantityMagnitude @ UnitConvert @ IsotopeData["Barium144 ", "AtomicMass "]) c ^ 2

Out[59]= 299 792 458

Out[60]=  $3.91962887 \times 10^{-25}$ 

Out[61]=  $3.52278675 \times 10^{-8}$ 

Out[62]=  $1.371922581 \times 10^{-8}$ 

Out[63]=  $2.14793218 \times 10^{-8}$ 
```

```
In[67]:= (* Calculate energy released from one fission. *)
Efission = E236 - E092 - E144
e = QuantityMagnitude @ UnitConvert @ Quantity @ "ElectronCharge ";
(Efission / e) 10 ^(-6)
```

```
Out[67]= 2.93199 × 10-11
```

```
Out[69]= 183.000
```

```
In[70]:= (* Calculate the fission rate and burnup time. *)
rate = 500 * 10 ^6 / Efission
```

```
Out[70]= 1.70533 × 1019
```

```
In[71]:= time = (10 / m236) / rate;
time / (24 * 60 * 60)
```

```
Out[72]= 17.3154
```

Example 9.3

The mass of the Sun is much larger than any of the planets. Assuming circular orbits, estimate the solar mass from data for each planet. Plot the ratio of this estimate to the accepted value of the solar mass, as a function of distance of the planet to the Sun.

```
In[72]:= (* MASS OF THE SUN FROM PLANET DATA *)
(* Get the planet data. *)
planets =
{"Mercury", "Venus", "Earth", "Mars", "Jupiter", "Saturn", "Uranus", "Neptune"};
PlanetData["Properties"];
planetDistances =
QuantityMagnitude @ UnitConvert @ PlanetData[planets, "AverageOrbitDistance "]
planetPeriods = QuantityMagnitude @ UnitConvert @ PlanetData[planets, "OrbitPeriod "]
```

```
Out[74]= {5.9133491 × 1010, 1.0821141 × 1011, 1.49618773 × 1011, 2.28931109 × 1011,
7.79323489 × 1011, 1.42881720 × 1012, 2.874165879 × 1012, 4.498418710 × 1012}
```

```
Out[75]= {7.600544 × 106, 1.9414149 × 107, 3.1558149 × 107, 5.9355036 × 107,
3.7435566 × 108, 9.2929236 × 108, 2.6513700 × 109, 5.2004186 × 109}
```

```

In[76]:= (* Derive solar mass. *)
G = QuantityMagnitude @ UnitConvert @ Quantity @ "GravitationalConstant "
v = 2 Pi planetDistances / planetPeriods
solarMasses = planetDistances v^2 / G

Out[76]= 6.674 × 10-11

Out[77]= {48 884.22 , 35 021.485 , 29 788.898 ,
          24 234.112 , 13 080.165 , 9660.602 , 6811.1643 , 5435.0237}

Out[78]= {2.1172 × 1030 , 1.9885 × 1030 , 1.9892 × 1030 , 2.0144 × 1030 ,
          1.9977 × 1030 , 1.9979 × 1030 , 1.9978 × 1030 , 1.9909 × 1030}


In[109]:= (* Plot relative to accepted value of solar mass. *)
solarMass = QuantityMagnitude @ UnitConvert @ Quantity @ "SolarMass "
points = Transpose[{Log10[planetDistances ], solarMasses / solarMass}]

(* Transpose has the effect of Riffle+Partition *)
ListPlot[points, PlotRange → {{10.5, 13}, {0.98, 1.08}},
  AxesLabel → {"Log10 Distance", "Calculated / True"},
  PlotMarkers → Style["●", Orange, Large],
  GridLines → {None, {{1, Red}}}, PlotLabel → "Solar mass Estimate/Actual "
]

Out[109]= 1.988 × 1030

Out[110]= {{10.77183352 , 1.065}, {11.03427305 , 1.000},
           {11.174986090 , 1.000}, {11.359704813 , 1.013}, {11.891717766 , 1.005},
           {12.15497667 , 1.005}, {12.4585118293 , 1.005}, {12.6530598768 , 1.001}}

```



Chapter 10 Random Numbers

Example 10.1

A large number of ants are located at one point. Each takes a 1 mm long step in a random direction.

Then, from their new position, they each take another 1 mm step in a random direction. The process repeats. How many steps does it take for half of the ants to leave a circle with radius 1 cm?

```
In[19]:= (* ANTS EXECUTING A RANDOM WALK *)
nAnts = 1000;
nSteps = 200;
stepSize = 1;
radius = 10;
(* Calculate the walk. *)
 $\theta$  = RandomReal[{0, 2 Pi}, {nSteps, nAnts}];
xSteps = stepSize Cos[ $\theta$ ];
ySteps = stepSize Sin[ $\theta$ ];
xWalk = Accumulate[xSteps];
yWalk = Accumulate[ySteps];


In[28]:= (* Plot the walk of the first ant. *)
xVals = Flatten[{0, Transpose[xWalk][[1]]}];
yVals = Flatten[{0, Transpose[xWalk][[2]]}];

antPath = Transpose[{xVals, yVals}];
```

```

In[31]:= Show[
  ListPlot[antPath, Joined → True, PlotMarkers → {Style["●", Small, Orange]},
    (*to get the circle, use "[Filled Circle]" without space. *)
    PlotRange → {{-20, 20}, {-20, 20}}, AspectRatio → 1, PlotStyle → Orange ],
  ParametricPlot[{radius Cos[u], radius Sin[u]}, {u, 0, 2 Pi}, PlotStyle → Red]
]


```



```

In[32]:= (* Analyze the collection of ants. *)
GraphicsRow[{
  ListPlot[Transpose[{xWalk[[nSteps / 4]], yWalk[[nSteps / 4]]}],
  PlotRange → {{-50, 50}, {-50, 50}},
  AspectRatio → 1, PlotLabel → "Positions after 50 steps",
  ListPlot[Transpose[{xWalk[[nSteps / 2]], yWalk[[nSteps / 2]]}],
  PlotRange → {{-50, 50}, {-50, 50}},
  AspectRatio → 1, PlotLabel → "Positions after 100 steps",
  ListPlot[Transpose[{xWalk[[nSteps / 1]], yWalk[[nSteps / 1]]}],
  PlotRange → {{-50, 50}, {-50, 50}},
  AspectRatio → 1, PlotLabel → "Positions after 200 steps"}],
ImageSize →
Full]


```




```

In[59]:= rWalk = Sqrt[xWalk ^ 2 + yWalk ^ 2];
histLimit = 3 radius;
GraphicsRow[{
  Histogram[rWalk[[nSteps / 4]], {0, histLimit, 0.5}, PlotRange → {{0, 30}, {0, 70}}],
  Histogram[rWalk[[nSteps / 2]], {0, histLimit, 0.5}, PlotRange → {{0, 30}, {0, 70}}],
  Histogram[rWalk[[nSteps / 1]], {0, histLimit, 0.5}, PlotRange → {{0, 30}, {0, 70}}]
}, ImageSize → Full]

```



```
In[49]:= GraphicsRow[{
  ListPlot[Mean[Transpose[rWalk]]],
  ListPlot[StandardDeviation[Transpose[rWalk]]/Mean[Transpose[rWalk]]],
  ImageSize -> Automatic]
```



```
In[63]:= Length@Select[Map[Max, Transpose[rWalk]], # < radius &]
```

Out[63]= 98

```
In[64]:= Length@Select[rWalk[[nSteps]], # < radius &]
```

Out[64]= 410

Chapter 11 Animation

Example 11.1

Create an animation that shows two finite length wave trains, with the same wavelength, moving towards each other in opposite directions and at the same speed. Choose parameters that make it clear that when the two waves overlap, they generate a standing wave.

(* STANDING WAVE FROM TRAVELING WAVES *)


```
In[47]:= (* Form the traveling wave trains. *)
trainR = Sin[Pi (x - v t)/2] * UnitBox[(x - v t)/width];
trainL = Sin[Pi (x + v t)/2] * UnitBox[(x + v t)/width]; (*UnitBox aids
  visualization by showing the wave across only one unit of the abscissa. *)
trains = trainR + trainL
```

```
Out[49]= Sin[1/2 Pi (-t v + x)] * UnitBox[-t v + x/width] + Sin[1/2 Pi (t v + x)] * UnitBox[t v + x/width]
```

```
In[50]:= (* Set parameters. *)
v = 1;
end = 35;
width = 20;
tend = (end - 0.6 width) / v;
```



```
In[54]:= (* Create the animation. *)
Animate[Plot[trains /. t -> time, {x, -end, end},
PlotRange -> {{-end, end}, {-2, 2}},
{time, -tend, tend}, AnimationRunning -> False]
```



Example 11.2

An elliptical football is thrown at a 30° angle from the horizontal, with an initial speed of 25 m/sec.

Animate the motion of the football, including its tilt along the arc, assuming it was thrown in a “perfect spiral” so that the football’s nose is always in the direction of flight. Ignore air resistance.

```

In[15]:= (* THROWING AN AMERICAN FOOTBALL . *)
(* Define and derive necessary expressions . *)
x = v0 Cos[θ] t;
y = v0 Sin[θ] t - g t^2 / 2;

vx = D[x, t];
vy = D[y, t];
φ = ArcTan[vx, vy]
path = y /. Solve[xpath == x, t];
ymax = y /. Solve[vy == 0, t][[1]];
thitsol = Solve[y == 0, t][[2]];
thit = t /. thitsol;
range = x /. thitsol;
point = {x, y};

(* Initialize some parameters . *)
g = 9.8; v0 = 25; θ = 30 Pi / 180;


(* Produce the animation . *)
xlim = 1.1 range; ylim = 1.1 ymax;
Animate [Show[
Plot[path, {xpath, 0, range}, AspectRatio → ylim/xlim,
PlotRange → {{0, xlim}, {0, ylim}},
PlotStyle → Dotted, Axes → {True, False}],
Graphics[Rotate[
Circle[point /. t → time, {1.5, 1.0}](*ellipse*),
φ /. t → time]]],
{time, 0, thit}, AnimationRunning → False]

```

```

Out[19]= ArcTan[ $\frac{25\sqrt{3}}{2}$ ,  $\frac{25}{2} - 9.8 t$ ]

```



Example 11.3


Create a realistic animation of a swinging plane pendulum. Use the correct numerical solution and prove to yourself the motion looks correct when the initial angle is close to the top of the pendulum swing.

In[60]:=

```
(* PENDULUM ANIMATION *)
(* Set up and solve differential eqn. *)
phi0 = 99 Pi / 100;
tMax = 10;
eqn = phi'[t] + (2 Pi)^2 Sin[phi[t]] == 0;
sol = NDSolve[{eqn, phi[0] == phi0, phi'[0] == 0}, phi, {t, 0, tMax}];
```

In[64]:=


```
(* Plotting the solution. *)
Plot[phi[t] /. sol, {t, 0, tMax}, PlotStyle -> Thick]
```



```
In[65]:= (* Animate the solution. *)
len = 1;
rBob = 0.05 len;
xBob = len Sin[φ[t]] /. sol;
yBob = - len Cos[φ[t]] /. sol;
pBob = {xBob[[1]], yBob[[1]]};
```

```
In[70]:= Animate[Show[
Plot[0, {x, -len, len}, AspectRatio → 1],
Graphics[{Thick, Line[{0, 0}, pBob /. t → time]}],
Graphics[{Black, Disk[pBob /. t → time, rBob]}],
{time, 0, tMax},
AnimationRate → 0.5,
AnimationRunning → False]
```

Out[70]=



Chapter 12 Advanced Visualizations

Example 12.1

Create a three dimensional plot of the orbits of Jupiter, Saturn, Uranus, Neptune, and Pluto. Include a color scheme with labels.


```
In[16]:= (* ORBITS OF THE OUTER PLANETS *)
(* Retrieve the planet orbit data. *)
planets = {"Jupiter", "Saturn", "Uranus", "Neptune"};
orbits =
  Flatten[{PlanetData[planets, "OrbitPath"], MinorPlanetData["Pluto", "OrbitPath"]};
pluto2 = Flatten[{planets, "Pluto"}];
```

```

In[10]:= (* Plot the orbits. *)
colors = {Red, Blue, Green, Magenta, Orange};
GraphicsRow[Graphics3D[Table[{colors[[i]], orbits[[i]]}, {i, 1, 5}],
Column[
Table[{colors[[i]], pluto2[[i]]}, {i, 1, 5}]]]

```

Out[11]=



{Red, Jupiter}
 {Blue, Saturn}
 {Green, Uranus}
 {Magenta, Neptune}
 {Orange, Pluto}

Example 12.2

Two electric charges of opposite sign but equal magnitude are separated by some distance. (This is called an “electric dipole.”) Create a plot showing the field lines and equipotentials from such system, going out to distances several times the separation of the charges.

```

In[20]:= (* POTENTIAL AND FIELD OF AN ELECTRIC DIPOLE *)
(* Calculate for two charges on the z-axis. *)
d = 1;
rp = {x, y, z - d/2};
rn = {x, y, z + d/2};
v = 1 / Sqrt[rp.rp] - 1 / Sqrt[rn.rn];
e = -Grad[v, {x, y, z}];

```

```

In[32]:= (* Making a 2D plot on the xz-plane. *)
vv = v /. y -> 0;
ee = Part[e /. y -> 0, {1, 3}];
Show[
StreamPlot[ee, {x, -3, 3}, {z, -3, 3}],
ContourPlot[vv, {x, -3, 3}, {z, -3, 3},
Contours -> {-0.45, -0.25, -0.125, 0, 0.125, 0.25, 0.45},
ContourShading -> None],
Graphics[{Red, Disk[{0, d/2}, d/8],
White, Text["+", {0, d/2}]}],
Graphics[{Blue, Disk[{0, -d/2}, d/8],
White, Text["-", {0, -d/2}]}]]

```

Out[34]=

